
Abstract

Constraint Handling Rules (CHR) is an extension to Prolog which opens up a spec-
trum of hypothesis-based reasoning in logic programs without additional interpre-
tation overhead. Abduction with integrity constraints is one example of hypothesis-
based reasoning which can be implemented directly in Prolog and CHR with a
straightforward use of available and efficiently implemented facilities.

The present paper clarifies the semantic foundations for this way of doing abduc-
tion in CHR and Prolog as well as other examples of hypothesis-based reasoning
that is possible, including assumptive logic programming, hypotheses with priority
and scope, and nonmonotonic reasoning.

Examples are presented as executable code so the paper may also serve the addi-
tional purpose of a practical guide for developing such programs, and it is demon-
strated that the approach provides a seamless integration with existing constraint
solvers.

Key words: Abduction, Abduction as deduction, Hypothesis-based reasoning,
Logic programming

1 Introduction

Abduction in logic programming has proved to be a powerful technique for
solving a range of complex problems, and a large number of approaches have
been reported, see [1–4] for overview. Most known systems are based on meta-
level architectures which means that abductive logic programs are interpreted

by other logic programs rather than being compiled in an efficient way.

We see abduction as one example within a class of more general hypothesis-
based reasoning paradigms. The advantages of abduction are its straightfor-
ward declarative specification and the fact that a large class of relevant prob-
lems fits naturally into the framework: given an observation O, the task is
to find an abductive explanation A which is a set of hypotheses which, when
added to the current knowledge base, can explain O and does not introduce
inconsistencies. The observation O may be observed symptoms of a malfunc-
tioning system, and A a diagnosis for what may have caused O; or O can
represent a desired goal to be achieved and A a set of requirements that must
be present in order to reach that goal, e.g., a plan of primitive actions to be
performed. While abduction concerns the creation of single sets of hypotheses
that can explain the top query, there are other and more dynamic ways to
apply hypotheses during a computation, as will be shown below.

In the present paper, we consider general hypothesis-based reasoning realized
using available and efficiently implemented logic programming technology in a
direct way, which eliminates any interpretative overhead. More specifically, we
suggest prolog as an overall driver engine (and, as knowledge specification
language) with Constraint Handling Rules, chr, complementing the ability to
manage hypotheses by declarative specifications and efficient implementation.

Example 1.1 The following program, which specifies and solves an abductive
problem, is written in the syntax for chr in SICStus and swi prolog.
Constraint predicates are distinguished from normal prolog predicates by
declarations. Constraints of chr play the role of abducibles and integrity
constraints are written as chr rules, e.g., as indicated below with ‘==>’.

:- chr_constraint professor/1, rich/1, has_good_students/1.

professor(X),rich(X) ==> fail.

happy(X):- rich(X).

happy(X):- professor(X), has_good_students(X).

Execution of the query ?- professor(peter), happy(peter) results in a final
constraint store (= abducible explanation) consisting of {professor(peter),
has good students(peter)}. The chr rule excludes an explanation which
includes rich(peter), and thereby forcing the prolog engine to try the
second clause in its eagerness to verify the query. 2

The approach provides a seamless and efficient integration with all facilities of
the underlying prolog and chr system, including existing constraint solvers,
in a way that makes it possible to go beyond strictly abductive reasoning.
Futhermore, subtle problems with variables in abducible hypotheses in many
earlier approaches do not arise here. The main weakness of using chr and
prolog for abduction in this way, when comparing with some other systems,
is the limited use of negation.

By hypothesis-based reasoning, we refer to a space of problem solving and
programming techniques in which logic programs are extended with a global
state of hypotheses which interacts with the logic program and can be ma-
nipulated implicitly as in abductive reasoning or in more explicit ways, and
which may or may not relate in all details to declarative specifications. It will
be shown how abduction and other instances of hypothesis-based reasoning
can be realized in straightforward ways with chr and prolog, and we intend
in this way also to indicate that there is a rich scope for developers to produce
their own variants for different purposes.

Constraint Handling Rules were introduced in the early 1990es by Thom
Frühwirth (primary reference [5] from 1998 provides background and early
history), and is now available as extensions to major prolog systems, in-
cluding SICStus and swi. The idea of using chr for abduction was origi-
nally suggested by Slim Abdennadher and the present author in 2000 [6]. The
combination of prolog and chr for abductive and other kinds of hypothesis-
based reasoning has been developed together with Verónica Dahl since 2002
as a central collaborator, e.g., [7–9].

The present paper aims at giving a coherent presentation of the approach,
clarifying the semantic foundations (that were left implicit in earlier publica-
tions) as well as exposing hypothesis-based reasoning with prolog and chr
as a powerful and flexible programming paradigm.

Examples have been checked using SICStus prolog [10]; we use in most
cases the chr syntax and facilities provided by SICStus prolog version 4
which is intended to be identical to that of swi prolog [11]. Some extensions
to chr that we describe below are implemented using specifics of SICStus
prolog version 3, which differs in some details and has a larger collection of
low-level facilities, so transfer to version 4 may not be trivial in all cases.

2 Syntax and semantics of chr and its extensions

Where prolog represents a top-down, backward chaining computational para-
digm, Constraint Handling Rules, chr, extends with bottom-up, forward
chaining computations. Operationally, chr is defined as rewriting rules over a
constraint store, which can be seen as a global resource to be used by a pro-
log program for storing, manipulating and consulting different hypotheses.
This resource-oriented understanding of chr has led to the formulation of a
semantics for chr [12] based on linear logic (that we have not applied here).
We introduce firstly chr, then extend it with disjunctions into chr∨ [13], and
finally we combine it with prolog into a language which we refer to in this
paper as prolog+chr.

2.1 Constraint handling rules, chr

Chr inherits the basic nomenclature of prolog, and we refer to the notions
of constant and function symbols, predicates, variables, atoms, 1 terms and
queries and use initial capital letter for variables, etc. Substitutions, grounding
and renaming substitutions are defined as usual.

The predicates in a chr program are called constraint predicates, belonging to
disjoint sets of program specific ones and a fixed set of built-in predicates each
having a fixed meaning, including = with its standard meaning of syntactic
equality, and true and false; we assume a theory B for the built-ins, e.g., B |=
a = a and B 6|= false. A constraint is an atom with a constraint predicate,
which may be further classified as program specific or built-in according to
its predicate; in contexts with no ambiguity, ‘constraint’ may also be used for
‘constraint predicate’. A chr program consists of rules of the following kinds.

Simplification rules: h1, . . . , hn<=>Guard | b1, . . . , bm
Propagation rules: h1, . . . , hn==>Guard | b1, . . . , bm
Simpagation rules: h1, . . . , hk \ hk+1, . . . hn<=>Guard | b1, . . . , bm

Each hi is a program specific constraint, each bi is a program specific or built-
in constraint, and the guard G is a possibly empty sequence of built-in ones;
we have n,m ≥ 1 and n > k ≥ 2. The comma is a sequencing operator which
may be interpreted as conjunction or set construction. The constraints to the
left of the arrow symbol constitute the head 2 of the rule, those to right of the
vertical bar the body. For simplicity, we require that any variable in the guard
of a rule must occur also in its head. An empty guard is interpreted as true

and may be left out together with the vertical bar.

The rules of a chr program can be understood operationally as rewrite rules
over states which are sets of constraints. For the purpose of extending the
semantics to prolog below, we separate a state into two components as
〈Θ,Σ〉, where Θ is called the current query (also called unseen constraints),
and Σ the constraint store which does not contain equalities, as they are
treated in a special way. A state 〈∅,Σ〉 may be identified with Σ, and 〈Θ, ∅〉
with Θ when no ambiguity occurs. A distinguished state is denoted ⊥ and
understood as falsity or contradiction.

1 There is a slight confusion here. Some prolog sources use ‘atom’ to refer to
constants; we use ‘atom’ in the same way as the literature on mathematical logic,
i.e., predicate(term1, . . . , termn), n ≥ 0.
2 Primary sources on chr refer to each ci, i = 1..n as a head, and thus such rules as
multiheaded; complying with our own publications, we let ‘head’ refer to the whole
lefthand side.

A rule can apply if it has an instance so that 1) the head constraints become
identical to constraints in the store, and 2) the instantiated guard is satis-
fied. Simplification rules implement bi-implication by replacing a set of con-
straints in the store by other, equivalent constraints, while propagation rules
implement implication by adding new constraints without removing their ‘an-
tecedents’. Simpagation rules are, as their name suggests, a mix of the two
others: head constraints indicated before the backslash stay in the store and
those after are removed. Equality atoms s = t are executed as in Prolog by ap-
plying a most general unifier of s and t to the state. These principles define an
abstract derivation relation over states, denoted ;; we refer to [5] for details.
A more detailed, operational semantics, which is deterministic and captures
the semantics of the implemented chr systems, can be found in [14].

A query Q is given as a conjunction of constraints. A (general) derivation for
Q, given a chr program C, is given as a sequence of the steps above using
rules of C,

〈Q, ∅〉 = S0 ; S1 ; S2 · · · .
When the derivation is finite and ends with a state S, we write

Q
C
; S.

If S = ⊥ we say that the derivation has failed ; otherwise, if no step can be
applied to S = 〈∅, A〉 = A, we say that the derivation has (successfully) termi-
nated and that A is a set of answer constraints for Q. The answer substitution
σ related to Q and A is defined as the composition of the most general unifiers
applied in the derivation of A restricted to the variables of Q; we may write

this Q
C
; A, σ.

Example 2.1 The following chr program is written in the syntax provided
by the libraries supplied with swi and SICStus prolog.

:- chr_constraint a/0, b/0.

a,b <=> true.

The query ?-a produces the answer constraint set {a} whereas ?-a,b yields ∅.
While this operational behaviour is perfectly sound with respect to the logic
semantics to be introduced, this example indicates also the power of chr to
work very explicitly with generating hypotheses and controlling their scope;
above, the call to a makes the awareness of hypothesis a globally available
whereas the effect of b in the extended query is effectively to erase a. 2

As opposed to prolog, we need for chr only a single derivation to char-
acterize the set of all answers to a given query. Notice that chr in practice
applies committed choice which means that no alternatives can be obtained
backtracking, and a failure in one step means failure of the entire execution.

In practice, chr applies a multiset semantics and it is often desirable to add
additional rules to a chr program that suppress duplicates. The detailed
operational semantics may be employed by the programmer to obtain effects
that are otherwise difficult to achieve within an abstract, nondeterministic
semantics. In the few cases we employ such tricks below, the procedural details
are explained.

A chr program C is called failure-safe if, for any Q with C |= ∀(Q↔ ⊥), any
derivation leads to the state ⊥. A chr program is called a constraint solver
if it is failure-safe and any derivation terminates. The stronger property of
confluence is not needed for our results and sometimes not even desirable;
it means that any derivation for a given query leads to equivalent answers
independently of which alternative derivation is chosen; see [5].

Example 2.2 The following chr program defines a constraint solver.

:- chr_constraint here/1, there/1.

here(X), there(X) ==> fail.

It returns ⊥ whenever a query implies that some value v is both here(v) and
there(v). For queries without this property and without built-ins, it returns
the query as answer. 2

Example 2.3 A useful built-in is the dif/2 predicate provided by SICStus
prolog, which serves as a lazy non-equality test. Consider the constraint
dif(s,t): if s and t become non-unifiable, the constraint vanishes; when they
become identical, a failure is produced; otherwise it remains silent. When
an infinity of constant symbols are assumed, any finite set of reduced dif

constraints is satisfiable, i.e., each of the form dif(s,t) where s and t are
unifiable and non-identical. 2

The rules of chr are specific logic formulas written in a notation adapted for
computers with implicit quantifiers. The translation is given as follows,

Propagation rule Simplification rule

CHR rule: H==>G|B H<=>G|B

Logical meaning: ∀x̄(G→ (H → ∃z̄B)) ∀x̄(G→ (H ↔ ∃z̄B))

where x̄ refers to the variables in H and z̄ to those in B not overlapping
with x̄. 3 Here we understand a simpagation rule H1 \ H2<=>G | B as an
abbreviation of the simplification rule H1, H2<=>G | H2, B and translate it into

3 Recall that we required, for simplicity only, that any variable in the guard appears
in the head as well.

a logical formula accordingly. From [5] we have the following results which are
straightforward to show by induction. For simplicity, we formulate correctness
conditions for ground queries only.

Theorem 2.1 (Soundness and completeness of chr derivations)
Let C be a chr program, Q a ground query and A an answer constraint set,

i.e., Q
C
; A. Then C,B |= ∀(Q↔ A).

Whenever C,B |= ∀(Q ↔ A′) for a ground query Q which has at least one
terminated derivation, there is an answer constraint set A such that C,B |=
∀(A↔ A′) and Q

C
; A. 2

While soundness is as one may expect, the completeness part is rather weak
in the sense that not every constraint store which is a logical consequence of
the program will be found.

Example 2.4 Let C consist of the rules a<=>b and c<=>b. We have that

a
C
; {b} and that C,B |= ∀(a ↔ b). On the other hand, it holds that C,B |=
∀(a↔ c), but it is not the case that a

C
; {c}. 2

This indicates that a chr programmer needs to be aware of the operational
semantics to have the desired answers produced. When a deterministic oper-
ational semantics is introduced, this is even more apparent.

The following example shows that the requirement of a finite derivation is
necessary in theorem 2.1.

Example 2.5 (From [5]) Consider the program consisting of the single rule
p<=>p. For the query p we have that C,B |= ∀(p↔ p), but there are no answer
constraint sets since no derivation terminates. 2

For chr programs of propagation rules only, derivations generate models. The
answer constraint set for a query Q is the smallest model of the program which
contains Q.

Example 2.6 Consider the following chr program which extends example 2.2.

:- chr_constraint here/1, there/1, distant/2.

here(X), there(X) ==> fail.

here(X), there(Y) ==> distant(X,Y).

The query ?- here(me), there(you) generates as answer the model {here(me),
there(you), distant(me,you)}, whereas ?- here(you), there(you) gener-
ates ⊥ meaning that there is no model containing those two atoms. 2

2.2 Chr∨: chr with disjunction

For historical and conceptual reasons, we introduce the language chr∨ [13]
which is an extension of chr with possible disjunctions in the body, denoted
by semicolons. An operational semantics for chr∨ is provided by extending
the ; relation with the principle of selecting nondeterministically one of the
candidates of a disjunction.

In practice, since disjunctions are interpreted by the underlying Prolog system,
the different alternatives are tried out under backtracking. In order to present
a completeness property, we define a chr∨ tree for a query Q and program
C as a tree which has state 〈Q, ∅〉 as root. A node N may have exactly one
subtree labeled N ′ whenever N ; N ′, except for disjunctions, where each
alternative gives rise to a subtree. A tree is final whenever it is finite and no
more subtrees can be added; a tree and the query at its root are said to be
failed, if every branch ends with ⊥.

The advantage of chr∨ over chr is that it provides a way to explore different
possible hypotheses as may be needed in abductive reasoning.

Example 2.7 Consider the following chr∨ program in which q can be inves-
tigated in two alternative ways.

:- chr_constraints q/0, a/0, b/0, c/0.

q <=> a ; c.

a, b <=> false.

The query ?-q leads to two possible answer constraint sets {a} and {c},
whereas ?-b,q has only one, namely {b, c}. 2

The declarative semantics of chr∨ is defined in the same way as for chr, with
semicolon read as ∨. The following results generalizes theorem 2.1.

Theorem 2.2 (Soundness and completeness of chr∨ derivations) Let
C be a chr∨ program, Q a ground query, and assume there is a final chr∨

tree which contains the answer constraint sets A1, . . . , An. Then it holds that

C,B |= ∀(Q↔ A1 ∨ · · · ∨ An)

and
C,B |= ∀(Ai → Q), for all i, 1 ≤ i ≤ n.

Whenever C,B |= ∀(A′ → Q) for a ground query Q which has at least one
terminated derivation, there is an answer constraint set A such that C,B |=
∀(A′ → A) and Q

C
; A. 2

Example 2.8 We consider an example from the logic programming folk-
lore. Let gw stand for grass is wet, r for rained last night, and s for
sprinkler was on. Consider first the following program C1 consisting of a
simplification rule.

:- chr_constraint gw/0, r/0, s/0.

gw <=> r ; s.

The query ?- gw leads to two final constraint sets {r} and {s}. We confirm
the theorem noting that C1 |= gw ↔ r ∨ s, and that C1 |= r → gw. We notice
that if, in the completeness part, we let Q = gw and A′ = {gw}, we can use
A = {r}.

Consider another program C2 in which the simplification rule is replaced by
a propagation rule gw ==> r ; s. Here we get the final constraint sets {gw,
r} and {gw, s}. Again, we confirm the theorem noting that C2 |= gw ↔
(gw ∧ r) ∨ (gw ∧ s) and that C1 |= gw ∧ r → gw. We notice that if, in the
completeness part, we let Q = gw and A′ = {gw}, we can use A = {gw, r}. 2

2.3 Prolog+chr

Another way to explore different possible hypotheses is to combine prolog
with chr, where the disjunction is provided by alternative clauses for the same
prolog predicate; we refer to this combined language as prolog+chr. A
prolog+chr program has constraint and built-in predicates as for chr and
additionally a set of prolog predicates, disjoint from the two others; an atom
with a prolog predicate is called a prolog atom. A prolog+chr program
〈P , C〉 has two components, a chr∨ program C and a finite set P of clauses
of the form

h :- b1, . . . , bn, n ≥ 1

where h is a prolog atom called the head of the clause, and b1, . . . ,bn called
the body may consist of prolog atoms and constraints. (In some of our ap-
plications, disjunctions in chr rule bodies may be useful as a way to have
a constraint solver produce alternative solutions, we do not need this in the
bodies of prolog rules). A clause of the form h:-true is called a fact and
written as h; any other clause is called a rule. In prolog+chr we allow also
calls to prolog predicates in the bodies of chr rules.

The operational semantics for chr is extended to prolog+chr in the stan-
dard way: when a rule is applied to a prolog atom, it is unified with the head
of the clause, leading to either failure, i.e., ⊥, or the addition of the atoms
from the clause body to the current query.

A prolog+chr tree for a query Q and program 〈P , C〉 is defined analogously
to chr∨ trees, with alternative subtrees corresponding to the possible clauses
that may apply. All other notions are defined as before.

The operational semantics of available prolog+chr systems uses the stan-
dard top-down, left-to-right, textual-order computation rule of prolog. When
a constraint c is encountered and a chr rule is called, its body B is executed
also in the top-down, left-to-right. When this B has executed, and if c has not
been consumed by a simplification or simpagation step, the system searches
for other chr rules that may be activated by c, and so forth. In case no
prolog predicates are called from the chr rules, this means than when a
constraint is called, the chr rules continue as long as possible and then give
back control to the calling prolog rule. When, furthermore, the chr part of
a prolog+chr program 〈P , C〉 satisfies the conditions for being a constraint
solver (failure-safe and always terminating), the programmer may consider C
as a black-box which tests whether c is compatible with the already accumu-
lated constraints; it may return the updated set of constraints in a ‘solved
form’, which means that execution continues along the given branch, or ⊥,
forcing prolog to backtrack.

Example 2.9 The following prolog+chr is written in the syntax provided
by swi and SICStus prolog. Constraint predicates are declared as in chr
above and prolog predicates such as q are given implicitly by their use as
usual.

constraints a/0, b/0, c/0.

q:- a.

q:- c.

a,b <=> false.

This program behaves in the same way as the chr∨ program of example 2.7
above, so for example ?-b,q leads to one answer constraint set {b, c}. 2

A general method for rewriting prolog programs into chr∨ programs was
given by [13], which generalizes trivially to prolog+chr; this has been illus-
trated by examples 2.7 and 2.9.

The advantages of using prolog+chr instead of chr∨ are more modular pro-
gram structures, a natural separation between backward and forward chaining
parts, and not least efficiency and acceptance in the logic programming soci-
ety. Furthermore, it provides access to use all facilities of a fully instrumented
prolog environment, including Definite Clause Grammars and various high
and low level auxiliaries.

There is an important difference in the ways the prolog and chr rules are
applied. Prolog uses unification between the head of a rule and an atom

in the state; this unification may specialize both and also affect other parts
of the state. Chr applies a different sort of matching: a specialization or an
instance of the rule is made such that the head constraints are found as copies
in the state; this does not in itself have any other side affects to the state.

Example 2.10 Consider the following prolog+chr program showing dif-
ferent sorts of rules.

:- chr_constraint c1/1, c2/1.

p1(a).

p2(X):- X=a.

c1(X) <=> X=a.

c2(a) <=> true.

Queries ?-p1(X) and ?-p2(X) yield answer substitution X=a. For ?-c1(X) and
?-c2(X), only the first one leads to X=a whereas the second returns the query
unaltered as answer constraint. 2

The logic semantics of a prolog program can be described by the so-called
Clark completion [15]; see also [16]. While each single clause stands for an
implication formula, the set of all clauses in a program with common head
predicate are combined into a bi-implication (‘only-if’).

For a program of prolog+chr, we take as its logic meaning the conjunction
of the completion of its prolog part and the meaning of its chr part as given
above. We combine standard results about prolog with theorem 2.1 into the
following.

Theorem 2.3 (Soundness and completeness of prolog+chr derivations)
Let 〈P , C〉 be a prolog+chr program, Q a ground query, and assume there is
a final prolog+chr tree which contains the answer constraint setsA1, . . . , An.
Then it holds that

P , C,B |= ∀(Q↔ A1 ∨ · · · ∨ An)

and

P , C,B |= ∀(Ai → Q), for all i, 1 ≤ i ≤ n.

Whenever C,B |= ∀(A′ → Q) for a ground query Q which has at least one
terminated derivation, there is an answer constraint set A such that C,B |=
∀(A′ → A) and Q

P,C
; A. 2

If we furthermore require that the chr part of a prolog+chr program is
a constraint solver, we can remove the requirement in the completeness part
about existence of a terminated derivation.

2.4 Additional facilities in current prolog+chr systems

Here we mention briefly facilities of current prolog and chr systems that are
useful for hypothesis-based reasoning and that are applied in the remainder
of this paper.

2.4.1 Syntactic and semantic extensibility

Prolog’s operator definitions combined with the so-called term expansion
facilities available in, among others, SICStus prolog are effective tools for
the programmer to provide a syntax and implementation for variations of the
logic programming paradigm.

We illustrate the principle here by an example, which should be sufficient for
understanding the applications later in this paper. For details and variations
between the prolog versions, we refer to [10,11].

Example 2.11 The chrg and hyprolog systems [17,18] both include a
where notation which may help to make complicated programs more readable.
So, for example, the clause p(X):- r(X,Y), z(Y,17), q(X) may be written
alternatively in the following way,

p(X):- Test, q(X)

where Test = ( r(X,Y), z(Y,17) ).

The where device is defined as an operator with suitable precedence and as-
sociativity in the usual prolog way, and its ‘meaning’, i.e., replacing the
variable Test by the term given after the where symbol, is defined by adding
a suitable clause to the system predicate term_expansion; details are straight-
forward and omitted. 2

2.4.2 Dynamic assertion and retraction of prolog clauses

Using prolog’s familiar asserta, assertz, and retract primitives, it is
possible to accumulate and manipulate information in a global state during
the execution of a prolog program. In some cases, e.g., when constraints
are known always to be ground, chr constraints can be rewritten through
these facilities, but the chr rules need to be rewritten in an awkward way. We
consider this program style as obsolete, but mention the possibility as a target
for automatic or systematic, manual translation when reasons of efficiency may
make it relevant. In section 7.3 below, we show details and compare efficiency
for selected examples.

2.4.3 Low-level chr features

The different versions of chr have different collections of low-level tools, some
of which are useful for implementation of new facilities for hypothesis-based
reasoning. Some of these may be hard to relate to a declarative semantics,
but applied in a structured way and hidden well inside abstractions, they
can provide a logical behaviour that goes beyond what is expressed naturally
in plain chr. Such facilities may include explicit inspection of the constraint
store, including adding and deleting constraints bypassing chr’s normal mode
of working. Finally, we mention the so-called passive declarations which can
be added to a chr rule, and which are useful for a more detailed control of
when rules are applied.

Example 2.12 Consider the following rule which is a simplification rule ex-
tended with a passive declaration.

a(X), b(X)#Id <=> c(x) pragma passive(Id).

This rule cannot fire due to a call to b/1, but a call to a/1 initiates a search
for a possible companion b/1 constraint. So the query ?-a(1),b(1) returns
the constraints unaltered as answer, whereas ?-b(1),a(1) returns {c(1)}. 2

Passive declarations can be used for optimization alone or for changing the
logic of the program (analogously to the way ‘!’ is used by prolog program-
mers); [19] analyzes the different usages of passive declarations.

3 Abductive logic programming with chr∨ and prolog+chr

3.1 Abductive logic programs

Here we give a definition of Abductive Logic Programming adapted from [9].
An abductive logic program [2] is a triplet 〈P ,A, IC〉 where P is a logic
program, A a set of abducible predicates that do not occur in the head of any
clause of P , and IC a set of integrity constraints assumed to be consistent.
A (not necessarily ground) atom with an abducible predicate is called an
abducible (atom). We assume additionally that P and IC can refer to a set
of built-in predicates that have a fixed meaning given by the theory B. Given
abducible predicates A and integrity constraints IC, we define for a set of
abducibles and built-in atoms A ∪ I, a consistent ground instance to be a
common ground instance A′ ∪ I ′ of A ∪ I so that

• B |= I ′ (the instance of built-ins is satisfied)
• B ∪ A′ |= IC (the instance of abducibles respects the integrity constraints)

For simplicity and without loss of generality, we consider only ground queries;
an abductive explanation for a query Q and abductive logic program 〈P ,A, IC〉
is a set of abducibles and built-in atoms A ∪ I such that

• A ∪ I has at least one consistent ground instance A′ ∪ I ′,
• for any such A′ ∪ I ′, we have P ∪ A′ |= Q.

3.2 Prolog+chr programs as abductive logic programs

The following theorem shows the relationship between prolog+chr pro-
grams and a specific form of abductive logic programs. The proof follows
immediately from theorem 2.3 and the classical deduction theorem, which
states that Γ, γ |= φ iff Γ |= γ → φ; see, e.g., [20] or another good book on
mathematical logic.

Theorem 3.1 Let 〈P ,A, IC〉 be an abductive logic program that coincides
with a prolog+chr program 〈P , IC〉 withA being the set of program defined
constraints, and assume that IC satisfies the conditions for being a constraint
solver. Then any answer constraint set for a ground query Q to 〈P , IC〉 is
an abductive explanation for Q in the abductive program 〈P ,A, IC〉. For any
abductive answer A′ for a ground query Q, there is an answer constraint set
A (for Q in 〈P , IC〉) such that IC,B |= ∀(A′ → A). 2

A similar theorem holds for chr∨ with the prolog part of the abductive logic
program rewritten, via its completion, to a set of simplification rules.

Example 3.1 (Adapted from [18]) We consider a planning problem in con-
struction work, concerned with the building of a facade of a classical temple.
The architect’s sketch is as follows, naming the different parts and showing
their position in the desired construction.

soil
f0
f1

c1 c2

gable

The following program expresses the problem as an abductive logic program
written in prolog+chr. We provide the entire source file, including the di-
rective to include the chr library, so the text can be copied and run directly
in SICStus prolog.

:- use_module(library(chr)).
:- chr_constraint put_in_place/2. % abducibles put_in_place(Object,Time)

%%% A part cannot be placed earlier that a part that supports it:
put_in_place(P0,Time0), put_in_place(P1,Time1) ==>

supports(P0,P1), Time0 > Time1 | fail.

build:- put_in_place(soil,0), parts(Parts), build(Parts,1).

build([],_).

build(Parts,Time):-
takePart(P,Parts,RestParts),
put_in_place(P,Time),
Time1 is Time+1,
build(RestParts,Time1).

parts([gable,c1,c2,f0,f1]).
supports(soil,f0).
supports(f0,f1).
supports(f1,c1).
supports(f1,c2).
supports(c1,gable).
supports(c2,gable).

takePart(X,List0,List1):-
append(LeftRest,[X|RightRest],List0),
append(LeftRest, RightRest,List1).

The following dialogue shows the query and an example of an abductive answer
expressing a plan, i.e., a specification of which part of the construction should
be placed at which times.

?- build.

put_in_place(gable,5),

put_in_place(c2,4),

put_in_place(c1,3),

put_in_place(f1,2),

put_in_place(f0,1),

put_in_place(soil,0) ?

2

The teaching note [21] provides further examples, including diagnosis of faulty
logical circuits, where small adjustments of the integrity constraints can tune
the program to work under different assumptions of periodic faults, consistent
faults, and that observed correct behaviour always is produced in a correct
way, i.e., excluding the option that different faults compensate for each other.

Example 3.2 The dining philosophers problem was introduced in [22] as a
prototype problem for process synchronisation and also considered in the lit-
erature on abduction. We use here a formulation inspired by [23]. Consider
(here) five philosophers sitting at a round table with one chopstick placed
between each two philosophers. In order to eat, a philosopher needs two chop-
sticks which he can take from the table and but back afterwards. Clearly, no
two neigbouring philosophers can eat at the same time. We use the following
constraint solver to indicate the overall conditions.

:- chr_constraint chop_in_use/3, % chop_in_use(ChopId, PhiId, Time)

chop_free/2, % chop_free(ChopId, Time)

eating/3. % record who ate when:

% eating(PhiId, Start, End)

chop_in_use(C,Ph,Tx), chop_free(C,Ty)

==> true | (Tx=Ty ; dif(Tx,Ty) ).

chop_in_use(C,Phx,T), chop_in_use(C,Phy,T) ==> Phx=Phy.

Th first integrity constraint allows to take a chopstick which is free at time t
to be used by a philosopher starting from t; the second one states that only
one philosopher can use a given chopstick at any given time.

The following prolog facts describe the positioning around the table and a
simple implementation of time, indicating possible successive ‘eating intervals’.

compute_needed_chops(ph1,c1,c2). eat(t0,t1).

compute_needed_chops(ph2,c2,c3). eat(t1,t2).

compute_needed_chops(ph3,c3,c4). eat(t2,t3).

compute_needed_chops(ph4,c4,c5). eat(t3,t4).

compute_needed_chops(ph5,c5,c1). eat(t4,t5).

eat(t5,t6).

The sequence of actions performed by each philosopher is described by the
following prolog rule.

phil(P):-

compute_needed_chops(P,C1,C2),

chop_in_use(C1,P,T1), chop_in_use(C2,P,T1),

eat(T1,T2), eating(P,T1,T2),

chop_free(C1,T2), chop_free(C2,T2).

For simplicity, it is assumed that each philosopher wants to eat just once.
Now the following query defines the initial setup at time t0 and poses the
philosophers’ desire to eat.

?- chop_free(c1,t0), chop_free(c2,t0), chop_free(c3,t0),

chop_free(c4,t0), chop_free(c5,t0),

phil(ph1), phil(ph2), phil(ph3), phil(ph4), phil(ph5).

A number of solutions are produced on backtracking; here we show the first
and fifth.

eating(ph5,t4,t5), eating(ph5,t1,t2),

eating(ph4,t3,t4), eating(ph4,t0,t1),

eating(ph3,t2,t3), eating(ph3,t2,t3),

eating(ph2,t1,t2), eating(ph2,t1,t2),

eating(ph1,t0,t1) eating(ph1,t0,t1)

Adding more integrity constraints, we can ensure optimal solutions (as the
indicated fifth) in the sense that chopsticks are taken up if possible at any
given time. 2

The following example shows linguistic discourse analysis by means of abduc-
tion.

Example 3.3 (Adapted from [24]) Abduction is an appreciated metaphor
for discourse analysis from natural language texts; see, e.g., [25]. Chr works
smoothly together with prolog’s Definite Clause Grammar notation and,
although this is by no means a profound discovery, it provides great advantages
in at least the teaching of computational linguistics. Consider the following
discourse.

Garfield eats Mickey, Tom eats Jerry, Jerry was a mouse, Tom is a cat,
Mickey was a mouse.

A discourse analysis may reveal to which categories the mentioned characters
belong and which categories are food items for which others. A particularly
interesting question is to which category Garfield belongs as this is not men-
tioned explicitly. The following combination of chr and grammar rules im-
plements the necessary parsing and abductive reasoning framework to do the
job. Notice that the chr rules, i.e., the integrity constraints, are simpagation
rules; this removes the duplicate constraints that otherwise would arise with
a propagation rule after the unification in the body. The integrity constraints
indicate that the category of a given character is unique, and for the sake of
this example it is assumed that a given category is the food item for at most
one other category.

:- use_module(library(chr)).

:- chr_constraint categ_of/2, food_for/2.

categ_of(N,C1) \ categ_of(N,C2) <=> C1=C2.

food_for(C,C1) \ food_for(C,C2) <=> C1=C2.

sentences --> [] ; sentence, sentences.

sentence --> name(N), ([is];[was]), category(C), {categ_of(N,C)}.

sentence --> name(N1), [eats], name(N2),

{categ_of(N1,C1), categ_of(N2,C2), food_for(C2,C1)}.

name(N) --> [N].

category(C) --> [a], noun(C).

noun(N) --> [N].

The analysis of the discourse above yields an abductive answer corresponding
to a knowledge base learned from that discourse, which includes among others
the abducible categ_of(garfield,cat). A detailed trace shows that the first
sentence produces

categ_of(garfield,X1), categ_of(mickey,X2), food_for(X1,X2)

where the Xs are currently uninstantiated variables. As the subsequent sen-
tences are analyzed and more abducibles are added to the constraints, the
integrity constraints will graduately fire so that the final sentence leads to a
unification of variable X1 with the value cat. 2

3.3 Abductive logic programming with constraints

Application of existing constraint solvers may provide effective ways of reduc-
ing the search space in abductive reasoning and may also help in producing
more concise programs. We refer to such constraints as external constraints,
and the applied constraint solver as being external. 4 Systems for abductive
logic programming, which incorporate specific constraint solvers such as finite
domain solvers or clp(R), have been developed [26–28].

Any constraint solver provided by the given prolog system can be used in an
abductive logic program written in prolog+chr. It works correctly without
any special care from the programmer, since each constraint solver applies it
own constraint store, which becomes part of the total execution state. When a

4 This completes the slightly confusing overloading of the term “constraint”. Con-
straints may now refer to CHR constraints, taking the role of abducibles, integrity
constraints which are CHR rules, and now external ones which are prolog atoms
whose meaning is provided by the prolog system or one of its libraries.

prolog program takes care of the overall control, each constraint solver reacts
each time one of “its” constraints are called, performs possible transformations
in its own store, producing either a failure or a signal to the prolog engine
to continue with the next subgoal. However, this holds only when we do not
use prolog’s or other kinds of negation as failure, which then would require
an explicit manipulation of the constraint representation. We show here an
example which uses a clp(Q) constraint solver.

Example 3.4 We consider the construction of paths in a graph, whose lengths
are subject to restrictions. This example involves nonground abducibles and,
for the prolog+chr implementation, the use of an external constraint solver,
which here is SICStus’ library(clpq) [29,10].

goal

start

n1_1

n1_2 n2_2

n2_1 nn_2

nn_2

n1_1

n1_n n2_n

n2_1 nn_1

nn_n

 . . .

 . . .

 . . .

 . . .

 . . .

 . . .

 . . .

Nodes are labelled, start, goal, ni i. Edges, except in the top and bottom,
are bidirectional and assigned a random cost. The task is to find three different
paths, that we call the red, green and blue path, and which must be disjoint.
Furthermore, the collected cost for the red paths must be larger than the one
for the green path, successively larger than for the blue path.

Path costs can be calculated incrementally (which is the optimal way), but
in order to have the external constraint solver do interesting work, we set
up constraints when generating each path (from the top and downwards),
so the total and intermediates costs are not known before the path reaches
node start. However, the program may indirectly compare the costs of paths
generated from the cost of the edges already involved and the knowledge the
the unknown costs are positive. The graph is given by program facts as follows,
here shown for the instance n = 5.

node(start).
edge(start,n1_1,53).
...
node(n1_1).
edge(n1_1,n2_1,18).
edge(n2_1,n1_1,57).
edge(n1_1,n1_2,89).
edge(n1_2,n1_1,83).
...
node(n5_5).
edge(n5_5,goal,16).
node(goal).

The remaining program clauses are as follows, with path edge being the ab-
ducible predicate. Code inside the curly brackets is managed by clp(Q).

problem:-
{RedCost > GreenCost, GreenCost > BlueCost},
path(red,RedCost), path(green,GreenCost), path(blue,BlueCost).

path(Id,Cost):- path(Id,goal,Cost).
path(_,start,StartCost):- {StartCost=0}.
path(Id,N2,C):- edge(N1,N2,C2), {C = C1+C2, C1>=0},

path_edge(Id,N1,N2,C), path(Id,N1,C1).

Integrity constraints to prevent loops in paths and ensure disjointness are as
follows. 5

path_edge(Id,_,N,_), path_edge(Id,_,N,_) ==> fail.
path_edge(Id1,_,N,_), path_edge(Id2,_,N,_) ==>

Id1 \= Id2, N\=goal | fail.

The query ?- problem produces the following set of abducibles; path edge is
abbreviated pe and colours to single letters.

pe(g,n2_5,goal,264)
pe(r,n1_5,goal,287) pe(g,n2_4,n2_5,193) pe(b,n4_5,goal,230)
pe(r,n1_4,n1_5,205) pe(g,n2_3,n2_4,116) pe(b,n4_4,n4_5,221)
pe(r,n1_3,n1_4,203) pe(g,n2_2,n2_3,97) pe(b,n4_3,n4_4,128)
pe(r,n1_2,n1_3,156) pe(g,n3_2,n2_2,31) pe(b,n4_2,n4_3,108)
pe(r,n1_1,n1_2,142) pe(g,n3_1,n3_2,13) pe(b,n4_1,n4_2,65)
pe(r,start,n1_1,53) pe(g,start,n3_1,7) pe(b,start,n4_1,45)

2

5 Current implementations of chr will never apply a rule such as the first integrity
constraint in the path example with the two head literals matching the same con-
straint. Other systems may need to test that the start nodes are different.

3.4 Extensions: compaction and explicit negation

Several other systems for abductive logic programming include the principle
that we call compaction, which that, whenever a new abducible atom appears
during the execution, the system tries to unify it with existing abducibles if
possible. The aim of this is to produce minimal explanations measured in the
number of literals. As we have argued elsewhere [9], it is not always desirable
to apply this principle.

Example 3.5 Consider the following example of an abducible explanation
which indicates that some events e1 and e2 have taken place at some locations
and points in time.

event(e1,P1,T1), place(P1), time(T1),

event(e2,P2,T2), place(P2), time(T2)

According to the compaction principle, the place and time constraints can
be unified, and the explanation shrinks from 6 to 4 abducibles. However, this
adds the extra commitment stating that the two events took place at the same
place and the same time, which the problem or its specification very likely may
not account for. 2

The implementation of abductive logic programs as prolog+chr programs
described here does not provide compaction. However, if compaction is desired
for a given predicate, it can be implemented by adding a propagation rule. For
example 3.5 above, the following rules will equate places and time points as
much as possible without violating other integrity constraints.

place(P1), place(P2) ==> true | (P1=P2 ; dif(X,Y)).

time(T1), time(T2) ==> true | (T1=T2 ; dif(X,Y)).

The dif constraint is a SICStus buit-in explained in example 2.3 above. (It
is necessary to include the true guard due to a subtlety concerned with the
precedence of the applied operators.) Notice that due to the multiset semantics
in current chr systems, compaction rules may lead to duplicate constraints
in the store which may not be desirable; they can be removed, if wanted,
with rules such as place(P)\place(P)<=>true which will be triggered by the
unification and undone on backtracking.

Explicit negation is a form of negation in logic programming based on the
axiom p ∧ ¬p → false. We can implement this with a propagation rule; for
an abducible p (with, say two arguments), we may let p be a predicate that
represents ¬p, and add the following rule.

p(X,Y), p_(X,Y) ==> false.

Example 3.6 Consider the following program, extended with compaction
and explicit negation rules.

past:- a(1), b(2), b_(1).

obs(X):- a(X),b(X).

The query ?-past,obs(X) produces two answers on backtracking, first X=2,
secondly a set of constraints on X, {dif(X,1), dif(X,2), a(X), b(X)}. 2

3.5 Providing a specialized syntax for abductive logic programming

The hyprolog system [9,18] adds a thin layer of syntactic sugar on top
of SICStus prolog. Abducibles are declared by the syntax illustrated as
follows.

abducibles p/2, q/1.

Using term expansion facilities (section 2.4.1), this is immediately translated
into declarations of chr constraints 6 for the mentioned predicates and their
negations, plus the rules that implement explicit negation.

:- chr_constraint p/2, p_/2, q/1, q_/1.

p(X,Y), p_(X,Y) ==> false.

q(X), q_(X) ==> false.

Compaction is declared by a directive such as

compaction q/1.

which translates into a rule as follows.

q(X), q(Y) ==> true | (X=Y ; dif(X,Y)).

In fact, hyprolog produces a slightly different rule which applies low-level
chr facilities to remove duplicate constraints in a backtrackable way with no
additional rules of the kind indicated above in section 3.4.

Hyprolog supports other sorts of hypotheses-bases reasoning that are ex-
plained later.

6 Hyprolog is implemented in SICStus prolog 3; the syntax shown here is valid
only in SICStus 4, which we use for consistency throughout this paper.

4 Assumptions in logic programming

We consider here assumptions in logic programming in the sense of [30]; we
use a revised syntax introduced in [9]. Assumptions are related to abduction
in the sense that they represent hypotheses in a global state that are generated
and can be employed during the execution of a program.

As opposed to abduction, the meaning of these assumptions depends on the
sequential execution of a prolog program, such that the availability of certain
hypotheses may be delimited by points in time according to the execution
history. A continuation semantics for prolog with assumptions is given by [9];
here we will do with an informal presentation.

Given a set of assumption symbols, which is a set of prolog function symbols,
an assumption can be any term whose top symbol is an assumption symbol.
The following operators that operate on assumptions can be applied in the
body of prolog rules.

+A Assert linear assumption A for subsequent proof steps.
Linear means “can be used once”.

*A Assert intuitionistic assumption A for subsequent proof steps.
Intuitionistic means “can be used any number of times”.

-A Expectation: consume/apply existing intuitionistic assumption
in the state which unifies with A.

=+A, =*A, =-A
Timeless versions of the above, meaning that order of assertion
of assumptions and their application or consumption can be
arbitrary.

When different hypothesis are available for -A and =-A, they may all be tried
under backtracking.

Example 4.1 Assuming that h is an assumption symbol, the following query,

?- +h(1), -h(X), +h(2), +h(3), -h(Y).

yields two possible answers, X=1, Y=3, with assumption +(h(2)) remaining in
the state, and X=1, Y=2, with +(h(3)) remaining in the state. 2

It seems quite obvious that it should be possible to implement these assump-
tion operators in chr using the constraint store for maintaining the collection
of assumptions. However, the backtracking among alternative assumptions,
represented as constraints, does not fit with chr’s commited choice. This
problem can be solved, e.g., by keeping all available assumptions in a list
stored as a single constraint. We exemplify here the implementation of + and
-; the full collection of operators can be implemented in a similar way.

Example 4.2 Linear assumptions and expectations can be implemented by
the following lines of code.

:- chr_constraint (-)/1, (+)/1, assump_list/1.

+A, assump_list(L) <=> assump_list([A|L]).

+A <=> assump_list([A]).

-E, assump_list(L) <=> member(A,L,LRest), assump_list(LRest), A=E.

The predicate member/3 takes out an element of a list and returns the list of
the remaining elements; it generates all possible members on backtracking. In
provides the behaviour shown in example 4.1 above. 2

The hyprolog system [9,18], whose implementation of abduction was out-
lined in section 3.5 above includes also an implementation of assumptions with
the full collection of operators shown above. Assumption symbols are declared
explicitly, exemplified as follows.

assumptions h/1.

timeless_assumptions n/2.

The system translates these into declarations of constraints specialized for ef-
ficiency for each assumption symbol,’+h’/1, ’*h’/1, ’-h’/1 and ’=+n’/2,
’=*n’/2, ’=-n’/2, and specialized rules similar to those shown in exam-
ple 4.2. 7

Assumptions in logic programming have been used in language analysis to
model dependencies across tree structures such as pronoun resolution and
coordination sentences. The following example is adapted from [9]; see also [18]
for more elaborate linguistic examples.

Example 4.3 Consider discourses consisting of sentences such as “Peter likes
Mary. She likes him.” The following program which runs under the hyprolog
system, is written using prolog’s grammar notation and uses abducibles to
record the facts embedded in each sentence and assumptions in order to resolve
pronouns.

assumptions acting/2.

abducibles fact/3.

sentence --> np(A,_), verb(V), np(B,_), {fact(A,V,B)}.

sentences --> [] ; sentence(S1),sentences(S2).

np(X,Gender) --> name(X,Gender), {*acting(X,Gender)}.

7 The currently available hyprolog system, version 0.0, applies low-level chr
primitives available in SICStus prolog 3 only, but the cleaner implementation
shown in example 4.2 could have been used instead.

name(peter,masc) --> [peter].

...

np(X,Gender) --> {-acting(X,Gender)}, pronoun(Gender).

pronoun(fem) --> [her].

...

verb(like) --> [likes].

...

The rule for np(· · ·) --> name(· · ·), . . . produces an assumption *acting(X,

Gender) to indicate that the given name is available for any future pronoun;
when a pronoun is encountered, an expectation -acting(X,Gender) is issued
which then, can be unified with the possible assumptions in the state. In
the sample discourse above, *acting(peter,masc) and *acting(mary,fem)

are generated for the first sentence, so that the expectation *acting(X,fem)

generated for “She” will unify this X with mary. Thus the following abducibles
are generated for the sample discourse, fact(peter,like,mary), fact(mary,
like,peter). 2

Assumptions have also be used for simulation of resource scheduling; see [18]
for an example.

5 An example of hypotheses with priority and scope

Chr lacks facilities for a detailed priority among rules, and hence also for the
applications we have shown so far of hypothesis-based reasoning. As the fol-
lowing example indicates, it is possible to assign priority numbers to different
hypotheses, and then select the one with the highest number.

Example 5.1 The application of chr for analysis of use case descriptions
for object oriented systems development and converting them into UML di-
agrams has been considered in [31,32]. The following rule resolves pronoun
reference according to a heuristic of taking the most recent, relevant refer-
ent. 8 It is adapted from [31] with slightly enhanced readability using the
where notation introduced in example 2.11 above. Constraints referent/4

and expect referent/3 are used similarly to the assumptions and expecta-
tions described in section 4 above; arguments No can be plural or singular,

8 It should be stressed that pronoun resolution, and anaphora resolution in general,
is one of the most difficult tasks in computational linguistics; the interested reader
is referred to [33] for an insightful and entertaining exposition of some of the phe-
nomena that should be taken into account. The simplistic heuristic chosen in the
present example is only acceptable here as the texts in question are expected to be
highly stereotypical.

G stands for gender, and Id and X identify given referents, e.g., peter. Refer-
ents are numbered by the sentence number (in the given discourse) from which
priorities are calculated. When, say, “Peter” is mentioned in sentence no. 7, a
constraint referent(sing,masc,peter,7) is emitted, and an occurrence of
“him” in sentence no. 10 gives rise to expect referent(sing, masc,X); the
following rule attempts to bind X to the suitable value.

sentence_no(Now), referent(No,G,Id,T) \ expect_referent(No,G,X) <=>

T < Now, \+ A_more_recent_referent_possible

| ( An_equally_good_referent_possible

-> X = error:pronoun:ambiguous(No,G,Now)

; X=Id )

where

A_more_recent_referent_possible = (

( find_constraint( referent(No,G,_,TMoreRecent) ),

T < TMoreRecent, TMoreRecent\==Now) ),

An_equally_good_referent_possible = (

find_constraint( referent(No,G,Id1,T) ), Id1\=Id ).

The current sentence number is kept in the constraint sentence no(Now). A
given referent can only be chosen if it is the most recent one with the right
feature values. The test in the body leads to an error code in case there are
two equally and most recent possibilities; this means that the heuristic gives
up in cases like “Peter and Paul ... . He ...” The predicate find constraint is
a primitive available in swi and SICStus prolog which unifies its argument
with an existing constraint if possible (with different syntax and different
degree of documentation in the different prolog versions). 2

This example indicates also an obvious need for a higher-level notation to ex-
press such priorities and selection criteria. Proposals have been made for ex-
pressing priorities in logic program, which may be used in the present context,
e.g. [34–36]. We mention also recent work suggesting priorities in chr [37].

6 An example of nonmonotonic reasoning in CHR

The paper [38] describes an implementation in chr of a language for agent-
oriented programming language called Global Abduction introduced in [39,40].
The idea in Global Abduction is basically that one or more agents work on
solving a problem and when doing this, they maintain a common knowledge
base reflecting the state of a changing world. Each branch of the computation
keeps a record R of which beliefs it has applied from the common knowledge
base CB in its proof steps; as soon as R becomes inconsistent with CB due
to a nonmonotonic update of CB, it delays (perhaps indefinitely) until CB

and R become consistent; then it may take up its computations. In this way
knowledge learned about the world in one branch becomes available to other
branches, and ‘old’, partial solutions to the given problem may be reused when
they fit the current word.

The global knowledge base is represented by constraints (+)/1 and (-)/1

that hold information known to be true or false, respectively; notice that this
usage of plus and minus is totally unrelated to the use of these symbols for
the assumptions in logic programs considered in section 4.

Nonmonotonicity means that when a new belief, say +a, is added to the
database, it overrides a possible old belief about the opposite, i.e., -a. The
management of the global knowledge base is given by the following rules, the
two last ones removing duplicates in a trivial way.

constraints (+)/1, (-)/1.

+X \ -X#Old <=> true pragma passive(Old).

-X \ +X#Old <=> true pragma passive(Old).

+X \ +X <=> true.

-X \ -X <=> true.

If the knowledge base contains -a and process calls +a, then the first rule,
which is a simpagation, fires and thereby removes -a and replaces it by +a;
the passive declarations ensures that it is always the currently called constraint
that takes precedence.

Since Global Abduction delays a process that does not fit the global knowledge
base and may restart it when this is not the case anymore, there is no need
for complicated updates of each process state. The necessary delay and restart
mechanisms are described in [38].

7 Evaluation and comparison with related work

The language of prolog+chr (or chr∨ for that matter) is purely deductive
in the sense that it produces only answers that are logical consequences of the
program at hand. Referring to the deduction theorem, we twisted the correct-
ness statement (theorem 2.3) into a specification of abductive problem solving,
and thereby obtaining an implementation. This principle is also inherent in a
paper by Console et al [41] from 1991, that also explained abductive reasoning
in terms of deduction; however, they did not suggest an underlying implemen-
tation platform with the desired deductive capabilities. The idea is extended
into a procedure for abduction described as an abstract algorithm by Fung and

Kowalski in 1997 [42], which inspired several implemented systems. See [1–4]
for overview and detailed references to a plethora of different approaches and
implementations. The main point of our work in comparison, indicates that
we have identified a match of technology invented and implemented efficiently
by others, namely prolog+chr, and applied it in the approach to abduction
by deduction.

The price to be paid is the lack of any interesting use of negation, only the very
limited form of so-called explicit negation can be used; we used notation a

to represent negation of a. For example, with a rule a(X), b(X) ==> false

we do not automatically deduce b (1) from a(1); if we want this, the rule
should be rewritten as a(X) ==> b (X) and b(X) ==> a (X) assuming rules
for explicit negation as described in section 3.4. However, for rules that are
more complicated that this, such a rewriting becomes very difficult.

As another example of hypothesis-based reasoning in chr, [43] shows how
description logic can be expressed in chr. The demoII system [44,45] uses a
combination of chr and prolog to implement a powerful reasoning system
based on a reversible version of the classical metainterpreter demo(program,

query) for logic programs with the meaning that the query is provable in the
program; reversibility means the system will try to instantiate metavariables
in the program argument standing for unknown program components in order
to make the query provable. Applications includes abduction, default logic
and simplified induction problems. The paper [46] analyzes prolog and chr
as general metaprogramming languages.

Chr has also be applied in different ways for language analysis. We have
shown above, how hypothesis-based reasoning such as abduction and assump-
tion implemented in chr works smoothly together with prolog’s grammar
notation; [8] has related this to a possible worlds semantics and linguistic
theories about context comprehension. The chr grammar system [17,24],
or chrg for short, translates a grammar notation into chr rules that parse
a text bottom-up, and which also interacts with various sorts of hypotheses.
Other applications of chr for language processing that are not commented on
here may be found in [7,47–52].

Finally we refer to the website for Constraint Handling Rules which maintains
references to all known papers and applications of chr [53].

7.1 Comparison with other selected systems

Here we describe a number of comparable approaches for which implemen-
tations are available; a table of benchmarks is provided for some of them,
section 7.3 below.

• Implementation in prolog+chr as described in section 3.2 and also with
an external clp(Q) constraint solvers.
• Prolog with assert-retract, section 2.4.2, which for programs with guar-

anteed ground abducibles may provide high efficiency.
• A straightforward meta-interpreter implementation of abductive logic pro-

grams with ground abducibles. We adapted a version described in [54,55].
• Abductive logic programming systems extended with finite domain con-

straints. We consider ciff [28] and the Asystem [26]. Both implementa-
tions involve meta-interpretation, and both have concentrated on efficient
employment of SICStus prolog’s finite domain solver. This solver ba-
sically stores the constraints without any reduction or attempts to detect
failures, until the user program explicitly calls an instantiation phase, called
labeling. Ciff applies a strategy of testing if a labeling is possible from time
to time (i.e., not for every computation step) to ensure consistency of the
collected finite domain constraints; Asystem performs incremental consis-
tency checking and applies a strategy for selecting the next subgoal which
looks at the current domain for constrained variables.
• For comparison with the above, we test implementations in prolog us-

ing the finite domain solver without involving any abductive machinery or
special strategy for tuning the performance of the constraint solver.
• Answer set programming (ASP) systems, which compute stable models

bottom-up, and in which some abductive problems with integrity constraints
can be expressed. DLV [56] is known to be an efficient, state-of-the-art ASP
system, but we made no tests as DLV’s lack of functions symbols and con-
straint solving made it unfit with the test suite introduced below.

We used instead the Smodels system [57] as it handles function symbols.
In the processing of a query, this system applies first a grounding preprocess-
ing phase which produces representations of all possible ground instance of
the clauses, including integrity constraints, of the program. It is concluded
below, that this is prohibitive for a large class of abductive problems, as
the number of these instances grows exponentially with the number of vari-
ables in each clause. Ideas for combining ASP with constraint solving are
emerging [58,59] but we have not tested any running systems.
• BinProlog [60] includes a hardwired implementation of assumptions; an

earlier test [9] indicates that assumptions executes about 3 times faster in
BinProlog than in prolog+chr (not shown in the test table below).

7.2 A suite of test programs

Example 7.1 (Sample program: db update) Database update through
views are typical examples used for demonstrating abductive systems. We
use an example with three tables and a view predicate w/3 which joins all
tables. Each table has a key constraints, shown as propagation rules below.

:- chr_constraint abd_p/4, abd_q/4, abd_r/3.

w(E,F):- pp(A,B,C,D), qq(C,D,E,F), rr(A,E,F).

pp(A,B,C,D):- p(A,B,C,D).

pp(A,B,C,D):- abd_p(A,B,C,D).

qq(C,D,E,F):- q(C,D,E,F).

qq(C,D,E,F):- abd_q(C,D,E,F).

rr(A,E,F):- r(A,E,F).

rr(A,E,F):- abd_r(A,E,F).

abd_p(A,B,_,_) ==> \+ p(A,B,_,_). % A,B key

abd_q(C,D,_,_) ==> \+ q(C,D,_,_). % C,D key

abd_r(A,E,_) ==> \+ r(A,E,_). % A,E key

p(a1,b1,c1,d1). ... p(a100,b100,c100,d100).

q(c1,d1,e1,f1). ... q(c100,d100,e100,f100).

r(a1,e1,f1). ... r(a99,e99,f99).

The test query ?- w(e100,f100) results in the abducible abd r(a100, e100,

f100); notice that an implementation trying out clauses in textual order will
need to go through all tables many times in order to find the right place
to insert a new fact. Abducibles in the store are always ground, which means
that a translation is possible into prolog using assert-retract and hand-coded
integrity constraints. Ground abducibles are also a requirement for the use of
prolog’s negation as failure above to work correctly. This program is also
tested with Asystem, ciff, Smodels and demoII.

The integrity constraints can be written in a more elegant way in other sys-
tems, which allows non-abducible predicates in the left hand side of integrity
constraints; here we show the ciff version.

[abd_p(A,B,_,_), p(A,B,_,_)] implies [false]. % A,B key

[abd_q(C,D,_,_), q(C,D,_,_)] implies [false]. % C,D key

[abd_r(A,E,_), r(A,E,_)] implies [false]. % A,E key

Overall, this example is characterized by a large base of facts which is accessed
very frequently (and where implementations that employ prolog’s indexing
are expected to be faster), lot of abducibles being generated, with integrity
constraints based on term (non-)equality, and plenty of failed alternatives.
However, there are at most two abducibles in existence at that same time. 2

Example 7.2 (Sample program: build(n)) This is an extension of the plan-
ning problem of example 3.1 to build houses of cards of height n, as illustrated
here for n = 2.

floor

card(1,1,r)card(1,1,l)

card(1,1,b)

card(2,1,r)

card(2,1,l)

card(2,1,b)

card(2,2,r)card(2,2,l)

card(2,2,b)

The top floor has number 1, and successively downwards and each “triangle”
is numbered 1, 2, ... from left to right in each floor. Each card is represented by
a term card(floor,triangle,x) where x is one of l(eft), r(ight) and b(ottom).

The build(n) program has O(n2) facts describing which cards support others,
and the task is to produce a plan for building the house, cf. example 3.1
above. We have extended the problem with an additional integrity constraint
so that the two slanting cards touching each other in the top must be placed
at successive moments. In CHR, the integrity constraints are as follows.

put_in_place(P0,Time0), put_in_place(P1,Time1) ==>
supports(P0,P1), Time0 > Time1 | fail.

put_in_place(card(N,L,l),Time0), put_in_place(card(N,L,r),Time1) ==>
1 is abs(Time1-Time0).

Cards are selected successively from a list which has been ordered in a non-
optimal way so that a very large number of wrong choices need to be ruled
out by the integrity constraints. Abducibles in the store are always ground,
which means that a translation is possible into Prolog using assert-retract
and hand-coded integrity constraints; this can be done as follows.

:- dynamic dyna_put_in_place/2.

put_in_place(C,T):- dyna_put_in_place(C,T), !.

put_in_place(C,T):-
( supports(C,C1), dyna_put_in_place(C1,T1), T > T1 -> fail ; true ),
( supports(C1,C), dyna_put_in_place(C1,T1), T1 > T -> fail ; true ),
( C=card(N,L,LorR), dyna_put_in_place(card(N,L,LorR1),T1),

LorR\==b, LorR1\==b, 1 =\= abs(T-T1)
-> fail ; true),

( assert((dyna_put_in_place(C,T)))
; retract((dyna_put_in_place(C,_))) ).

The ciff integrity constraints are encoded in the following way.

[put_in_place(P0,Time0), put_in_place(P1,Time1),
supports(P0,P1), Time0 #> Time1] implies [false].

[put_in_place(card(N,L,l),Time0), put_in_place(card(N,L,r),Time1),
(Time1-Time0)*(Time0-Time1) #\= -1] implies [false].

Overall, this example is characterized by a large base of facts with a lot of
abducibles being generated, with integrity constraints involving integer arith-
metic, and multitudes of failed alternatives. 2

Example 7.3 (Sample program: graph(n)) This program is introduced
in example 3.4 above. The version for prolog+chr uses SICStus’ clpq

library. The program can be rewritten directly into the syntax of ciff, which
handles the constraints using SICStus’ finite domain solver. We test only
for small graphs, so this example is characterized by small bases of facts, but
with potentially many possibilities that need to be tried out. Integrity checking
depends on an external constraint solver. 2

Example 7.4 (Sample program: nqueens(n)) This is adapted from an
example program for solving the classical n-queens problem which has been
used in papers about the ciff system [28] and the Asystem [26] to demon-
strate an effective interaction with an external constraint solver, which is SIC-
Stus prolog’s finite domain solver as explained above. The ciff version is
as follows, here shown for n = 20; the version for Asystem is almost identical
in appearance.

abducible(q_pos(_,_)).
q_domain(R) :- R #>= 1, R #=< 20.
exists_q(R) :- q_pos(R,C),q_domain(C).
safe(R1,C1,R2,C2) :- C1#\=C2, R1+C1#\=R2+C2, C1-R1#\=C2-R2.
[q_pos(R1,C1),q_pos(R2,C2),R1#<R2] implies [safe(R1,C1,R2,C2)].
q:- exists_q(1), ..., exists_q(20).

This program essentially translates the problem into one set of constraints,
and the finite domain constraint solver’s labeling toolbox is applied in different
ways by different systems for solving the problem.

In the adaptation to prolog+chr, we run an abductive phase producing
the constraints followed by a labeling of all variables occurring in q pos/2

constraints (using the ffc option, cf. [10]). Finally, we compare with an im-
plementation written directly in prolog which produces an “optimal” set of
constraints without the symmetric and redundant constraints produced by the
other versions; then a labeling phase (using the ffc option) is initiated. We
include a number of tests for different values of n in order to indicate that the
time complexity is not as regular as it may appear from the referenced papers;
see also [61] for an analysis of this example. 2

7.3 Tests and benchmarks

All tests were made on a Macintosh Intel Core Duo, 2GHz, 2GB Ram run-
ning Mac OS X 10.4.11. Prolog+chr and Prolog with assert-retract tests
made with SICStus 4.0.1 (i386-darwin-8.9.1); Prolog with clpfd, Asystem
and ciff with SICStus 3.12.8 (i386-darwin-8.9.1); Smodels is a standalone
application written in C. Times are in milliseconds unless otherwise stated;
a dash indicates that a test is not applicable and “∞” that the query did
not terminate in observable time (1h or more). For tests involving SICStus
prolog, we measure CPU time excluding garbage collecting etc., as speci-
fied by its option statistics(runtime, ); see [10]. For Smodels, the total
execution is measured for both grounding and model construction.

As it appears in the table, there are many unexpected values and outliers,
so all tests have been checked several times, and those involving SICStus
prolog have been replicated in a Windows-based environment, showing an
identical distribution of relative times used.

db update(3) 0.43 0.034 4.8 - 13 170 600

build(10)(4) 340 10 50 - n/a(5) 530 ∞(8)

build(15)(4) 3350 100 195 - n/a(5) 2250 ∞(8)

graph(3)6 180 - - - n/a(5) 4020 ∞(8)

graph(4)(6) 2 - - - n/a(5) 1160 ∞(8)

graph(5)(6) 14 - - - n/a(5) 2880 ∞(8)

nqueens(20)(7) 35 - - 5 60 80 ∞(8,9)

nqueens(32)(7) 45 - - 10 180 220 ∞(8,9)

nqueens(64)(7) 1300 - - 94 990 920 ∞(8,9)

nqueens(99)(7) ∞ - - 31m 11m 12m ∞(8,9)

nqueens(100)(7) 450 - - 108 60 2390 ∞(8,9)

nqueens(101)(7) 37m - - 1m15s 36s 40s ∞(8,9)

Notes

(1) These versions can only handle ground abducibles.
(2) ciff has two modes of treating integrity constraints, with or without

grounding; the figures indicate the best setting in each case.
(3) As expected, prolog+chr runs considerably faster than meta-interpret-

ation-based systems, but the difference to the much faster prolog with
assert-retract is striking; the cost of using chr is a factor of 13. ciff and
Smodels are up to 18,000 times slower than the fastest one.

(4) As opposed to db update, the simple meta-interpreter runs faster for
the tested build(n) problems, which is a bit surprising. To explain this,
notice that the heads of the CHR rules for integrity constraints in the
prolog+chr version of db update contains only a single literal. This
means that CHR’s search for rules to apply for an abducible (CHR con-
straint) amounts more or less to a call to a prolog predicate, whereas
the build(n) problems have two head literals in their integrity rules, which
means that the constraint store must be searched for the second of those,
when a given call is executed. The similar search made by the simple
metainterpreter is done as sequential list search which is faster for small
constraint stores than chr’s more general search mechanism.

For the (again superior) assert-retract approach, the entering of an
abducible is a prolog call, and the search for companion abducibles
for checking integrity is also made as a prolog call, which means that
prolog’s inherent indexing is employed as far as possible.

(5) The Asystem works fine for the nqueens problems and db update, but
strange behaviour was observed for the build and graph problems. This
may be due to the fact that Asystem was developed in an earlier version
of SICStus prolog that was not available for our tests.

(6) The graph(n) problem does not become more complex with larger n as
there are many more paths to select from, and thus it is easier to find
three paths that satisfy the conditions.

ciff runs considerably slower that prolog+chr, which may be due
to the fact that different constraint solvers are used: the clp(Q) solver
applied with prolog+chr has inherently incremental consistency check-
ing, while ciff’s finite domain solver only finds inconsistencies when a
labeling phase is initiated.

(7) All tests of nqueens(n), except for Smodels, apply SICStus prolog’s
finite domain solver. We notice that the time does not grow monotoni-
cally with n, and there is no consistent correlation between the different
implementations. The actual time depends on the labeling strategy used
and whether it is applied incrementally for testing consistency, and it
seems that all the strategies tested have their “black holes”.

The running times indicate also that the difficulty in finding the first
solution depends on n in a tricky way; for example, the case n = 99 is the
most difficult one tested, while n = 100 is an especially easy case. The

prolog+chr implementation is clearly more efficient for small n due to
its direct execution, but the detailed control strategies for the constraint
solver applied in Asystem and ciff pay back for the larger values of n.

(8) Smodels applies a grounding preprocessing phase which enumerates
all possible ground instance of predicate calls. The number of such in-
stances grows exponentially with the number of variables in a clause. For
build(10), we have integrity constraints with four variables, each of which
can be shown to range over 166 different values, yielding around 8 ∗ 108

different instances which is prohibitive; a similar analysis can be made
for graph(n). Furthermore, the grounding phase excludes the use of lists
with standard predicates such as member and append.

(9) Smodels could produce a solution to nqueens(8); grounding took 0.17s
and model construction 0.14s, which is much slower than any of other
approaches tested.

8 Concluding remarks

While chr was originally conceived as a declarative programming language for
writing constraint solvers, it has proved to be suited for what we have called
hypothesis-based reasoning. We have taken a top-down approach, defining
high-level patterns such as abduction and shown how their semantics can be
realized by straightforward mappings into prolog with chr.

This provides generally a high efficiency when compared to implementations
based on meta-interpretation, especially when the problem at hand is domi-
nated by deductive steps (such as the db update problem, example 7.1 above)
in which case the major part of the work is executed directly by the prolog
engine, employing its optimizing compiler.

The main advantages of using prolog and chr in this way may be the very
flexible architecture which makes it possible to 1) implement high-level pat-
terns such as abduction in straightforward ways, and 2) “massage” such a
pattern to work with different sorts of knowledge bases (such as assumptions)
and integrate with existing constraint-solvers in a seamless way. On the neg-
ative side, we notice a limited support for negation when we insist on a direct
execution, and our tests indicate that some of our chr programs could run
considerably faster with better indexing techniques in the underlying engine.
However, any future improvement of chr concerning its efficiency and facil-
ities will be of benefit for hypothesis-based reasoning, whether programmed
directly in prolog and chr or compiled from high-level specifications into
such a platform.

35



Acknowledgement: This work is supported by the CONTROL project,
funded by Danish Natural Science Research Council.

References

[1] A. C. Kakas, R. A. Kowalski, F. Toni, Abductive logic programming, Journal
of Logic and Computation 2 (1993) 719–770.

[2] A. C. Kakas, R. A. Kowalski, F. Toni, The role of abduction in logic
programming, in: D. Gabbay, C. Hogger, J. Robinson (Eds.), Handbook of
Artificial Intelligence and Logic Programming, Vol. 5, Clarendon Press, 1998,
pp. 235–324.

[3] M. Denecker, A. C. Kakas (Eds.), Journal of Logic Programming. Special issue:
abductive logic programming, Vol. 44 (1-3), 2000.

[4] M. Denecker, A. C. Kakas, Abduction in logic programming, in: A. C. Kakas,
F. Sadri (Eds.), Computational Logic: Logic Programming and Beyond, Vol.
2407 of Lecture Notes in Computer Science, Springer, 2002, pp. 402–436.

